
BBaabb  1155  
MMeenngggguunnaakkaann  OOLLEE  DDrraagg--DDrroopp  
 

15.1. Apa Itu OLE Drag-Drop? 
 
OLE (Object Linking and Embedding) Drag-Drop adalah jenis drag-drop yang memungkinkan user 
untuk menyalin atau memindahkan data dari satu bagian ke bagian yang lain di dalam satu program atau 
dengan program yang lain. Sebagian besar program-program buatan Microsoft (seperti Microsoft Office) 
ataupun perusahaan lain (seperti Adobe PhotoShop) mendukung penggunaan OLE drag-drop. 
 
15.2. OLE Drag-Drop dengan VB 6.0 
 
VB 6.0 menyediakan beberapa property, method dan event yang berhubungan dengan OLE drag-drop. 
Operasi OLE drag-drop melibatkan object source dan object target. Hanya beberapa komponen pada  
VB 6.0 yang bisa digunakan sebagai object source maupun target, yaitu: TextBox, ComboBox, ListBox, 
Image, PictureBox, DirListBox dan FileListBox. Sedangkan komponen lainya hanya bisa digunakan 
sebagai object target, seperti: CommandButton, CheckBox, OptionButton, Label dan Form. 

 
Lisensi Dokumen: 
Copyright © 2005 IlmuKomputer.Com 
Seluruh dokumen di IlmuKomputer.Com dapat digunakan, dimodifikasi dan disebarkan 
secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus 
atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap 
dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin 
terlebih dahulu dari IlmuKomputer.Com. 

Property, method dan event yang berhubungan dengan OLE drag-drop adalah sebagai berikut: 
 

Property Object Keterangan 
OLEDragMode Source Menentukan bagaimana OLE drag dimulai: 

- Manual (vbOLEDragManual = 0, default) atau 
- Otomatis (vbOLEDragAutomatic = 1) 

OLEDropMode Target Menentukan bagaimana OLE drop dilakukan: 
- Diabaikan (vbOLEDropNone = 0, default) 
- Manual (vbOLEDropManual = 1) atau 
- Otomatis (vbOLEDropAutomatic = 2) 

Method Object Keterangan 
OLEDrag Source Digunakan untuk memulai OLE drag secara manual 

(property OLEDragMode diset vbOLEDragManual). 
Event Object Keterangan 
OLEStartDrag Source Saat operasi OLE drag-drop dimulai. 
OLECompleteDrag Source Saat proses drop pada object target selesai. 
OLEDragOver Target Saat pointer mouse melewati object target selama 

operasi OLE drag-drop berlangsung. 
OLEDragDrop Target Saat data di-drop pada object target. 

 
Ada tiga argumen penting pada event-event OLE drag-drop, yaitu: AllowedEffects, Effect dan Data. 
Argumen AllowedEffects dan Effect digunakan untuk menentukan efek OLE drag-drop yang 
diperbolehkan atau yang digunakan. Nilai untuk kedua argumen tersebut adalah sebagai berikut: 
 

Nilai Keterangan 
vbDropEffectNone = 0 Data tidak bisa diterima oleh object target. 
vbDropEffectCopy = 1 Data disalin ke object target. 
vbDropEffectMove = 
2 

Data dipindahkan ke object target. 

 
Sedangkan argumen data merupakan object DataObject yang digunakan selama operasi OLE drag-drop 
berlangsung. 
 
15.3. Object DataObject 
 
Untuk menangani OLE drag-drop secara manual digunakan object DataObject untuk menyimpan data 
dan format datanya. Property dan method yang dimiliki object DataObject adalah: 
 

Property Keterangan 
Files Koleksi nama-nama file (format datanya vbCFFiles) 
Method Keterangan 
Clear Menghapus semua data dan format datanya. 
GetData(format) Mengambil data yang tersimpan. Argumen format digunakan 

untuk menentukan format data yang diambil. 
GetFormat(format) Mengetahui format data yang tersimpan. Akan menghasilkan 

nilai True (bila formatnya sesuai) atau False (bila formatnya 
tidak sesuai). 

SetData value, [format] Menyimpan data dengan format tertentu. Argumen value 
adalah data yang akan disimpan. 

Nilai untuk argumen format adalah sebagai berikut: 
 

Nilai Keterangan 
Dikosongkan = 0 Format data akan ditentukan otomatis berdasarkan datanya. 
vbCFText = 1 Data teks 
vbCFBitmap = 2 Data bitmap (gambar) 
vbCFMetafile = 3 Data metafile (gambar) 
vbCFDIB = 8 Data device-independent bitmap (gambar) 
vbCFFiles = 15 Data file 

 
15.4. Contoh Program: OLE Drag-Drop Test 
 
1. Aktifkan program Microsoft Visual Basic 6.0. 
2. Bukalah kembali file project Latihan.vbp. 
3. Tambahkan form baru ke dalam project. 
4. Pada Jendela Object buatlah User Interface seperti ini: 

 

 
 
Setting property object-nya adalah sebagai berikut: 
 

Object Properties Value 
Form15 BorderStyle 

Caption 
1 – Fixed Single 
OLE Drag-Drop Test 

Text1 OLEDragMode 
OLEDropMode 
Text 

1 – Automatic 
2 – Automatic 
<kosong> 

Image1 OLEDragMode 
OLEDropMode 
Stretch 

1 – Automatic 
2 – Automatic 
True 

 
5. Simpanlah Form15 dengan nama file Lat15.frm. 

Catatan: 
Anda bisa men-download file vb6-latihan.zip. 

6. Klik menu Project > Project1 Properties lalu klik tab General. Gantilah Startup Object-nya menjadi 
Form15. 

7. Coba jalankan project-nya dan aktifkan program WordPad: 
 

 

Keterangan: 
1. Buatlah teks dan 

gambar pada program 
WordPad. 

2. Lakukan drag-drop teks 
dari WordPad ke Text1. 

3. Lakukan drag-drop 
gambar dari WordPad 
ke Image1. 

 
Catatan: 
- Drag-drop tanpa 

tombol Ctrl  
teks/gambar akan 
dipindahkan. 

- Drag-drop dengan 
tombol Ctrl  
teks/gambar akan 
disalin. 

 
8. Ubahlah property object-object berikut ini: 

 
Object Properties Value 

Text1 OLEDragMode 0 – Manual 
List1 OLEDropMode 1 – Manual 
Image1 OLEDragMode 0 – Manual 
Image2 OLEDropMode 

Stretch 
1 – Manual 
True 

 
9. Buka Jendela Code, lalu ketikkan kode programnya seperti pada Listing-15. 
10. Coba jalankan kembali project-nya: 

 

 

Keterangan: 
1. Lakukan drag-drop teks dari Text1 ke List1. 
2. Lakukan drag-drop gambar dari Image1 ke 

Image2. 
 
Catatan: 
- Drag-drop tanpa tombol Ctrl  teks/gambar 

akan dipindahkan. 
- Drag-drop dengan tombol Ctrl  

teks/gambar akan disalin. 

11. Aktifkan program Windows Explorer: 
 

 

Keterangan: 
1. Lakukan drag-drop file dari 

program Windows Explorer ke 
List1. 

2. Lakukan drag-drop file dari 
program Windows Explorer ke 
Image2. 

 
Catatan: 
Drag-drop ke List1 bisa lebih dari 1 
file (pada program Windows 
Explorer gunakan Shift+klik atau 
Ctrl+klik untuk memilih lebih dari 1 
file). 

 
*** 

Listing-15 
 
Dim FormatOK As Boolean 
 
'====================== 
'OLE Drag-Drop ke List1 
'====================== 
Private Sub Text1_MouseDown(Button As Integer, Shift As Integer, _ 
                            X As Single, Y As Single) 
    'Jika Text1 tdk kosong dan tombol kiri mouse ditekan 
    If Text1.Text <> "" And Button = vbLeftButton Then 
        Text1.OLEDrag   'Mulai operasi OLE drag-drop 
    End If 
End Sub 
 
Private Sub Text1_OLEStartDrag(Data As DataObject, AllowedEffects As Long) 
    'Simpan data dgn format text 
    Data.SetData Text1.Text, vbCFText 
    'Atur efek OLE drag-drop yg diperbolehkan 
    AllowedEffects = vbDropEffectCopy Or vbDropEffectMove 
End Sub 
 
Private Sub List1_OLEDragOver(Data As DataObject, Effect As Long, _ 
                              Button As Integer, Shift As Integer, _ 
                              X As Single, Y As Single, State As Integer) 
    'Cek format datanya 
    FormatOK = Data.GetFormat(vbCFText) Or _ 
               Data.GetFormat(vbCFFiles) 
     
    If FormatOK Then 
        'Cek apakah tombol Ctrl ditekan 
        If Shift = vbCtrlMask Then 
            Effect = vbDropEffectCopy   'Data disalin 
        Else 
            Effect = vbDropEffectMove   'Data dipindahkan 
        End If 
    Else 
        Effect = vbDropEffectNone       'Data diabaikan 
    End If 
End Sub 
 
Private Sub List1_OLEDragDrop(Data As DataObject, Effect As Long, _ 
                              Button As Integer, Shift As Integer, _ 
                              X As Single, Y As Single) 
    Dim i As Integer 
     
    'Cek format datanya 
    FormatOK = Data.GetFormat(vbCFText) Or _ 
               Data.GetFormat(vbCFFiles) 
     
    If FormatOK Then 
        'Cek apakah tombol Ctrl ditekan 
        If Shift = vbCtrlMask Then 
            Effect = vbDropEffectCopy   'Data disalin 
        Else 
            Effect = vbDropEffectMove   'Data dipindahkan 
        End If 
 
 
 



Kuliah Berseri IlmuKomputer.Com 
Copyright © 2005 IlmuKomputer.Com 

Cepat Mahir Visual Basic 6.0  Bab 15 - 7 
Krisnha D. Oktovhiana 

        'Tambahkan data yg tersimpan ke dalam List1 
        If Data.GetFormat(vbCFText) Then 
            List1.AddItem Data.GetData(vbCFText)    'Data text 
        End If 
        If Data.GetFormat(vbCFFiles) Then 
            List1.Clear 
            For i = 1 To Data.Files.Count 
                List1.AddItem Data.Files(i)         'Data file(s) 
            Next i 
        End If 
    Else 
        Effect = vbDropEffectNone       'Data diabaikan 
    End If 
End Sub 
 
Private Sub Text1_OLECompleteDrag(Effect As Long) 
    If Effect = vbDropEffectMove Then 
        Text1.Text = ""     'Kosongkan Text1 jika data dipindahkan 
    End If 
End Sub 
 
'======================= 
'OLE Drag-Drop ke Image2 
'======================= 
Private Sub Image1_MouseDown(Button As Integer, Shift As Integer, _ 
                             X As Single, Y As Single) 
    If Image1.Picture <> 0 And Button = vbLeftButton Then 
        Image1.OLEDrag 
    End If 
End Sub 
 
Private Sub Image1_OLEStartDrag(Data As DataObject, AllowedEffects As Long) 
    Data.SetData Image1.Picture 
    AllowedEffects = vbDropEffectCopy Or vbDropEffectMove 
End Sub 
 
Private Sub Image2_OLEDragOver(Data As DataObject, Effect As Long, _ 
                               Button As Integer, Shift As Integer, _ 
                               X As Single, Y As Single, State As Integer) 
    FormatOK = Data.GetFormat(vbCFBitmap) Or _ 
               Data.GetFormat(vbCFMetafile) Or _ 
               Data.GetFormat(vbCFDIB) Or _ 
               Data.GetFormat(vbCFFiles) 
     
    If FormatOK Then 
        If Shift = vbCtrlMask Then 
            Effect = vbDropEffectCopy 
        Else 
            Effect = vbDropEffectMove 
        End If 
    Else 
        Effect = vbDropEffectNone 
    End If 
End Sub 

Private Sub Image2_OLEDragDrop(Data As DataObject, Effect As Long, _ 
                               Button As Integer, Shift As Integer, _ 
                               X As Single, Y As Single) 
    FormatOK = Data.GetFormat(vbCFBitmap) Or _ 
               Data.GetFormat(vbCFMetafile) Or _ 
               Data.GetFormat(vbCFDIB) Or _ 
               Data.GetFormat(vbCFFiles) 
     
    If FormatOK Then 
        If Shift = vbCtrlMask Then 
            Effect = vbDropEffectCopy 
        Else 
            Effect = vbDropEffectMove 
        End If 
        If Data.GetFormat(vbCFBitmap) Then 
            Image2.Picture = Data.GetData(vbCFBitmap) 
        End If 
        If Data.GetFormat(vbCFMetafile) Then 
            Image2.Picture = Data.GetData(vbCFMetafile) 
        End If 
        If Data.GetFormat(vbCFDIB) Then 
            Image2.Picture = Data.GetData(vbCFDIB) 
        End If 
        If Data.GetFormat(vbCFFiles) Then 
            Image2.Picture = LoadPicture(Data.Files(1)) 
        End If 
    Else 
        Effect = vbDropEffectNone 
    End If 
End Sub 
 
Private Sub Image1_OLECompleteDrag(Effect As Long) 
    If Effect = vbDropEffectMove Then 
        Image1.Picture = Nothing 
    End If 
End Sub 




